# Copyright (c) 2009 Nicira Networks.
#
# Licensed under the Apache License, Version 2.0 (the "License");
# you may not use this file except in compliance with the License.
# You may obtain a copy of the License at:
#
#     http://www.apache.org/licenses/LICENSE-2.0
#
# Unless required by applicable law or agreed to in writing, software
# distributed under the License is distributed on an "AS IS" BASIS,
# WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
# See the License for the specific language governing permissions and
# limitations under the License.

use strict;
use warnings;

my %counters;
while (<>) {
    my ($counter) = /^\s*COVERAGE_(?:INC|ADD)\s*\(\s*([a-zA-Z_][a-zA-Z_0-9]*)/
      or next;
    push (@{$counters{$counter}}, "$ARGV:$.");
} continue {
    # This magic resets $. from one file to the next.  See "perldoc -f eof".
    close ARGV if eof;
}

print <<EOF;
#include "coverage-counters.h"
#include <stddef.h>
#include "coverage.h"
#include "util.h"

EOF

for my $counter (sort(keys(%counters))) {
    my $locations = join(', ', @{$counters{$counter}});
    print <<EOF;
/* $locations */
struct coverage_counter ${counter}_count = { "$counter", 0, 0 };

EOF
}
print "struct coverage_counter *coverage_counters[] = {\n";
print "    \&${_}_count,\n" foreach (sort(keys(%counters)));
print "};\n";
print "size_t coverage_n_counters = ARRAY_SIZE(coverage_counters);\n";
